

LANDMINE

In Landmine the player must traverse a 10 by 10 field of possible mine locations, from top left to bottom right. The program sets up an almost too easy first two or three attempts but the machine is programmed to place the mines randomly increasing the numbers of mines by 2 each time. Rapidly the player is faced with the near impossible task of negotiating 48, 50 and more mines.

The program does offer the player a little help, in that it issues a warning whenever the player is one and/or two moves away from being blown to smithereens.

In giving the reader (and eventual user), the bare outlines of this addictive game, a lot of scope opens up for the ambitious and inventive home programmer. In this quiet utility program, he can create a more attractive display and with the increased sound capabilities of the 64 develop an exciting range of explosive sounds and warnings.

RUNNING THE PROGRAM

After the usual feat of typing in the program and typing 'RUN', the screen will display the instructions, giving the most important instruction of how to move the player's character:

```
7 8 9
4   6
1 2 3
```

The player takes the number 5 position, and corresponding to the compass, may use the numbers to move in any direction. Hitting any key will renew the game.

PROGRAM STRUCTURE

For an easy view of the positioning of interesting lines in the listing, the following has been correlated for the user's convenience:

70	Array for storing grass and mines.
90 - 130	Sets up grass in all positions of the array.
150 - 190	Chooses random positions for mines.
200 - 210	Sets grass in top left, house in bottom right position.
220 - 410	Displays minefield.
420 - 450	Input move.
460 - 640	Increase or decrease x and y position according to move input.
660 - 760	Check for mines near man.
780 - 810	Displays minefield successfully crossed.
830 - 920	Displays mine stepped on.
930 - 950	Pause for keypress, increase number of mines.
1000 - 1080	Display routine.
1110 - 1230	Checks and gives warning of mines near to man.
1320 - 1530	Instructions.

```

10 REM *****
20 REM *GAMES PROGRAM - SEE IF YOU CAN
30 REM *SUCCESSFULLY CROSS THE MINEFIELD
40 REM *****
50 REM SET UP MINEFIELD IN ARRAYS
55 POKE53281,1
60 NM=10:REM NO MINES = 10 INITIALLY
70 DIM A$(15,16)
80 GOSUB 1310
90 FOR I=1 TO 15
100 FOR J=1 TO 15
110 A$(I,J)="."
120 NEXT J
130 NEXT I
140 REM SET UP MINES AT RANDOM
150 FOR I=1 TO NM
160 X%=RND(1)*15+1
170 Y%=RND(1)*15+1
180 A$(X%,Y%)="M"
190 NEXT I
200 A$(1,1)="."
210 A$(15,15)="."
220 PRINT"XXXXXXXXXXXXXXXXXXXXL A N D M I N E"
230 PRINT"XXXXXXXXXXXXXXXXXXXX"
240 PRINT"NO OF MINES = ";NM
250 REM PRINT MINEFIELD ON SCREEN
260 CL=15:NO=15:Z$="." : T$="H"
270 FOR RW=5 TO 19
280 GOSUB 960
290 NEXT RW
300 CL=14:NO=17:RW=4:Z$="X" : T$="H"
310 GOSUB 960
320 RW=20
330 GOSUB 960
340 CL=14:NO=17:RW=4:Z$="X" : T$="V"
350 GOSUB 960
360 CL=30
370 GOSUB 960
380 X%=1:Y%=1
390 GOTO 660
400 GOSUB 1240
410 PRINT"@";
420 GET GG$:IF GG$="" THEN 400
430 GG=VAL(GG$)
440 IF GG<1 OR GG>9 THEN 400
450 ON GG GOTO 460,480,500,520,540,550,570,590,610
460 XX%=X%-1:YY%=Y%+1:CR$="X"
470 GOTO 620
480 XX%=X%:YY%=Y%+1:CR$="Y"
490 GOTO 620
500 XX%=X%+1:YY%=Y%+1:CR$="X"

```

```

510 GOTO 620
520 XX%=X%-1:YY%=Y%:CR$="III"
530 GOTO 620
540 GOTO 400
550 XX%=X%+1:YY%=Y%:CR$="III"
560 GOTO 620
570 XX%=X%-1:YY%=Y%-1:CR$="VII"
580 GOTO 620
590 XX%=X%:YY%=Y%-1:CR$="J"
600 GOTO 620
610 XX%=X%+1:YY%=Y%-1:CR$="VII"
620 IFXX%<1 OR XX%>15 THEN 400
630 IFYY%<1 OR YY%>15 THEN 400
640 X%=XX%:Y%=YY%
650 REM NOW CHECK FOR MINES IN VICINITY
660 IF A$(X%,Y%)="M" THEN 820
670 A$(X%,Y%)="●"
680 IF X%=15 AND Y%=15 THEN 780
690 REM CHECK FOR 1 MOVE AWAY
700 MS%=1:GOSUB 1090
710 IF MNC>0 THEN 740
720 REM CHECK FOR 2 MOVES AWAY
730 MS%=2:GOSUB 1090
740 REM RESET CURSOR WITHIN MINEFIELD
750 GOSUB 1240
760 GOTO 400
770 REM MINEFIELD CROSSED SUCCESSFULLY
780 CL=1:NO=1:RW=23:T$="H"
790 Z$="MINEFIELD SUCCESSFULLY CROSSED"
800 GOSUB 960
810 GOTO 860
820 REM LANDMINE EXPLODES
830 CL=1:NO=1:RW=23:T$="H"
840 Z$="B A N G !!!!!!!"
845 Z$=Z$+"
850 GOSUB 960
860 A$(X%,Y%)="X"
870 FOR Y=1 TO 15
880 X%=1:Y%=Y:GOSUB 1240
890 FOR X=1 TO 15
900 PRINT A$(X,Y);
910 NEXT X
920 NEXT Y
930 GET GG$:IF GG$="" THEN 930
940 IFGG$<>"E"THENNM=NM+2:GOTO90
950 END
960 REM SUBROUTINE TO DRAW LINE ON SCREEN
970 REM CL=START COLUMN,RW=START ROW
980 REM NO=NO OF CHARS,T$= V(VERT),H(HORIZ)
990 REM Z$=TEXT TO PRINT
1000 PRINT"■"

```

```

1010 FOR I=1 TO CL-1:PRINT"┌";:NEXT I
1020 FOR I=1 TO RW-1:PRINT"┘";:NEXT I
1030 IF T$="H" THEN I$=""
1040 IF T$="V" THEN I$="┌┐"
1050 FOR I=1 TO NO
1060 PRINTZ$;I$;
1070 NEXT I
1080 RETURN
1090 REM CHECK FOR MINES IN VICINITY
1100 REM MS%=1 CLOSE SEARCH,MS%=2 WIDE SEARCH
1110 MN=0
1120 FOR I=X%-MS% TO X%+MS%
1130 FOR J=Y%-MS% TO Y%+MS%
1140 IF I<1 OR J<1 THEN 1170
1150 IF I>15 OR J>15 THEN 1170
1160 IF A$(I,J)="M" THEN MN=MN+1
1170 NEXT J
1180 NEXT I
1190 Z$=" "
1200 IF MN<=0 THEN 1210
1204 Z$=STR$(MN)+" MINES(S) "
1206 Z$=Z$+STR$(MS%)+ " MOVES AWAY"
1210 CL=1:NO=1:RW=23:T$="H"
1220 GOSUB 960
1230 RETURN
1240 REM CURSOR ADDRESS TO WITHIN MINEFIELD
1250 REM X%=X AXIS, Y%=Y AXIS
1260 PRINT"Ⓚ";
1270 FOR I=1 TO X%+14:PRINT"┌";:NEXT I
1280 FOR I=1 TO Y%+4:PRINT"┘";:NEXT I
1290 PRINT"┌┐";
1300 RETURN
1310 REM INTRODUCTION & INSTRUCTIONS FOLLOW
1320 PRINT"#####L A N D M I N E#####"
1330 PRINT"#####"
1340 PRINT"THE OBJECT OF THIS GAME IS TO"
1350 PRINT"TRAVERSE A MINEFIELD. YOUR MINE"
1360 PRINT"DETECTOR GIVES SOME WARNING"
1370 PRINT"OF NEARBY MINES. BUT CAN'T"
1380 PRINT"GIVE THEIR EXACT LOCATION."
1390 PRINT"SO BE CAREFUL OR :-"
1400 PRINT"Ⓚ B A N G !!! Ⓚ"
1410 PRINT"YOU START AT THE TOP LEFT CORNER"
1420 PRINT"AND FINISH (HOPEFULLY) AT THE"
1425 PRINT"BOTTOM RIGHT."
1430 PRINT"USE THE NUMERIC PAD TO MOVE,"
1440 PRINT"WITH YOUR POSITION BEING RELATED "
1450 PRINT"TO THE 5 KEY.Ⓚ"
1460 PRINT"      7 8 9
1470 PRINT"      \ /

```

```
1480 PRINT"          4-●-6
1490 PRINT"          /|\
1500 PRINT"          1 2 3
1510 PRINT"■PRESS ANY KEY WHEN READY.■"
1520 GET GG$:IF GG$="" THEN 1520
1530 RETURN
```